Rules for integrands of the form $\sin\left[a + b x + c x^2\right]^n$

1. $\int \sin\left[a + b x + c x^2\right] dx$

1: $\int \sin\left[a + b x + c x^2\right] dx$ when $b^2 - 4 a c == 0$

- **Derivation: Algebraic simplification**
- **Basis: If** $b^2 - 4 a c == 0$**, then** $a + b x + c x^2 == \frac{(b+2 c x)^2}{4 c}$
- **Rule: If** $b^2 - 4 a c == 0$**, then**

$$\int \sin\left[a + b x + c x^2\right] dx \longrightarrow \int \sin\left[\frac{(b + 2 c x)^2}{4 c}\right] dx$$

- **Program code:**

```
Int[Sin[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  Int[Sin[(b+2*c*x)^2/(4*c)],x] /;
FreeQ[{a,b,c},x] && EqQ[b^2-4*a*c,0]
```

```
Int[Cos[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  Int[Cos[(b+2*c*x)^2/(4*c)],x] /;
FreeQ[{a,b,c},x] && EqQ[b^2-4*a*c,0]
```

2: $\int \text{Sin}[a+b\,x+c\,x^2]\,dx$ when $b^2-4\,a\,c\neq 0$

- **Derivation: Algebraic expansion**
- **Basis:** $a+b\,x+c\,x^2 == \frac{(b+2\,c\,x)^2}{4\,c}-\frac{b^2-4\,a\,c}{4\,c}$
- **Basis:** $\text{Sin}[z-w] == \text{Cos}[w]\,\text{Sin}[z]-\text{Sin}[w]\,\text{Cos}[z]$
- **Rule: If** $b^2-4\,a\,c\neq 0$**, then**

$$\int \text{Sin}[a+b\,x+c\,x^2]\,dx \longrightarrow \text{Cos}\Big[\frac{b^2-4\,a\,c}{4\,c}\Big]\int \text{Sin}\Big[\frac{(b+2\,c\,x)^2}{4\,c}\Big]\,dx-\text{Sin}\Big[\frac{b^2-4\,a\,c}{4\,c}\Big]\int \text{Cos}\Big[\frac{(b+2\,c\,x)^2}{4\,c}\Big]\,dx$$

- **Program code:**

```
Int[Sin[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  Cos[(b^2-4*a*c)/(4*c)]*Int[Sin[(b+2*c*x)^2/(4*c)],x] -
  Sin[(b^2-4*a*c)/(4*c)]*Int[Cos[(b+2*c*x)^2/(4*c)],x] /;
FreeQ[{a,b,c},x] && NeQ[b^2-4*a*c,0]
```

```
Int[Cos[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  Cos[(b^2-4*a*c)/(4*c)]*Int[Cos[(b+2*c*x)^2/(4*c)],x] +
  Sin[(b^2-4*a*c)/(4*c)]*Int[Sin[(b+2*c*x)^2/(4*c)],x] /;
FreeQ[{a,b,c},x] && NeQ[b^2-4*a*c,0]
```

2: $\int \text{Sin}[a+b\,x+c\,x^2]^n\,dx$ when $n\in\mathbb{Z}\,\land\,n>1$

Derivation: Algebraic expansion

Rule: If $n\in\mathbb{Z}\,\land\,n>1$**, then**

$$\int \text{Sin}[a+b\,x+c\,x^2]^n\,dx \longrightarrow \int \text{TrigReduce}\big[\text{Sin}[a+b\,x+c\,x^2]^n\big]\,dx$$

Program code:

```
Int[Sin[a_.+b_.*x_+c_.*x_^2]^n_,x_Symbol] :=
  Int[ExpandTrigReduce[Sin[a+b*x+c*x^2]^n,x],x] /;
FreeQ[{a,b,c},x] && IGtQ[n,1]
```

```
Int[Cos[a_.+b_.*x_+c_.*x_^2]^n_,x_Symbol] :=
  Int[ExpandTrigReduce[Cos[a+b*x+c*x^2]^n,x],x] /;
FreeQ[{a,b,c},x] && IGtQ[n,1]
```

X: $\int \text{Sin}\left[a+b\,x+c\,x^2\right]^n \,dx$

Rule:

$$\int \text{Sin}\left[a+b\,x+c\,x^2\right]^n \,dx \longrightarrow \int \text{Sin}\left[a+b\,x+c\,x^2\right]^n \,dx$$

Program code:

```
Int[Sin[a_.+b_.*x_+c_.*x_^2]^n_.,x_Symbol] :=
  Unintegrable[Sin[a+b*x+c*x^2]^n,x] /;
FreeQ[{a,b,c,n},x]
```

```
Int[Cos[a_.+b_.*x_+c_.*x_^2]^n_.,x_Symbol] :=
  Unintegrable[Cos[a+b*x+c*x^2]^n,x] /;
FreeQ[{a,b,c,n},x]
```

N: $\int \text{Sin}[v]^n \,dx$ when $n \in \mathbb{Z}^+ \wedge v == a+b\,x+c\,x^2$

Derivation: Algebraic normalization

\- **Rule: If** $n \in \mathbb{Z}^+ \wedge v == a+b\,x+c\,x^2$**, then**

$$\int \text{Sin}[v]^n \,dx \longrightarrow \int \text{Sin}\left[a+b\,x+c\,x^2\right]^n \,dx$$

Program code:

```
Int[Sin[v_]^n_.,x_Symbol] :=
  Int[Sin[ExpandToSum[v,x]]^n,x] /;
IGtQ[n,0] && QuadraticQ[v,x] && Not[QuadraticMatchQ[v,x]]
```

```
Int[Cos[v_]^n_.,x_Symbol] :=
  Int[Cos[ExpandToSum[v,x]]^n,x] /;
IGtQ[n,0] && QuadraticQ[v,x] && Not[QuadraticMatchQ[v,x]]
```

Rules for integrands of the form $(d+e\,x)^m\,\text{Sin}\left[a+b\,x+c\,x^2\right]^n$

1. $\int (d+e\,x)^m\,\text{Sin}\left[a+b\,x+c\,x^2\right]\,dx$

   1. $\int (d+e\,x)^m\,\text{Sin}\left[a+b\,x+c\,x^2\right]\,dx$ when $2\,c\,d-b\,e==0$

      1: $\int (d+e\,x)\,\text{Sin}\left[a+b\,x+c\,x^2\right]\,dx$ when $2\,c\,d-b\,e==0$

Derivation: Inverted integration by parts with $m\to 1$

Rule: If $2\,c\,d-b\,e==0$, then

$$\int (d+e\,x)\,\text{Sin}\left[a+b\,x+c\,x^2\right]\,dx\;\longrightarrow\;-\frac{e\,\text{Cos}\left[a+b\,x+c\,x^2\right]}{2\,c}$$

Program code:

```
Int[(d_+e_.*x_)*Sin[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  -e*Cos[a+b*x+c*x^2]/(2*c) /;
FreeQ[{a,b,c,d,e},x] && EqQ[2*c*d-b*e,0]
```

```
Int[(d_+e_.*x_)*Cos[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  e*Sin[a+b*x+c*x^2]/(2*c) /;
FreeQ[{a,b,c,d,e},x] && EqQ[2*c*d-b*e,0]
```

2: $\int (d+e\,x)^m \operatorname{Sin}[a+b\,x+c\,x^2]\,dx$ when $2\,c\,d-b\,e=0\land m>1$

Derivation: Inverted integration by parts

Rule: If $2\,c\,d-b\,e=0\land m>1$, then

$$\int (d+e\,x)^m \operatorname{Sin}[a+b\,x+c\,x^2]\,dx \longrightarrow -\frac{e\,(d+e\,x)^{m-1}\operatorname{Cos}[a+b\,x+c\,x^2]}{2\,c}+\frac{e^2\,(m-1)}{2\,c}\int (d+e\,x)^{m-2}\operatorname{Cos}[a+b\,x+c\,x^2]\,dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*Sin[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  -e*(d+e*x)^(m-1)*Cos[a+b*x+c*x^2]/(2*c) +
  e^2*(m-1)/(2*c)*Int[(d+e*x)^(m-2)*Cos[a+b*x+c*x^2],x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[2*c*d-b*e,0] && GtQ[m,1]
```

```
Int[(d_.+e_.*x_)^m_*Cos[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  e*(d+e*x)^(m-1)*Sin[a+b*x+c*x^2]/(2*c) -
  e^2*(m-1)/(2*c)*Int[(d+e*x)^(m-2)*Sin[a+b*x+c*x^2],x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[2*c*d-b*e,0] && GtQ[m,1]
```

3: $\int (d+e\,x)^m \operatorname{Sin}[a+b\,x+c\,x^2]\,dx$ when $2\,c\,d-b\,e=0\land m<-1$

Derivation: Integration by parts

Basis: $(d+e\,x)^m = \partial_x \frac{(d+e\,x)^{m+1}}{e\,(m+1)}$

Basis: If $2\,c\,d-b\,e=0$, then $\partial_x \operatorname{Sin}[a+b\,x+c\,x^2] = \frac{2\,c}{e}\,(d+e\,x)\,\operatorname{Cos}[a+b\,x+c\,x^2]$

Rule: If $2\,c\,d-b\,e=0\land m<-1$, then

$$\int (d+e\,x)^m \operatorname{Sin}[a+b\,x+c\,x^2]\,dx \longrightarrow \frac{(d+e\,x)^{m+1}\operatorname{Sin}[a+b\,x+c\,x^2]}{e\,(m+1)}-\frac{2\,c}{e^2\,(m+1)}\int (d+e\,x)^{m+2}\operatorname{Cos}[a+b\,x+c\,x^2]\,dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*Sin[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  (d+e*x)^(m+1)*Sin[a+b*x+c*x^2]/(e*(m+1)) -
  2*c/(e^2*(m+1))*Int[(d+e*x)^(m+2)*Cos[a+b*x+c*x^2],x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[2*c*d-b*e,0] && LtQ[m,-1]
```

```
Int[(d_.+e_.*x_)^m_*Cos[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  (d+e*x)^(m+1)*Cos[a+b*x+c*x^2]/(e*(m+1)) +
  2*c/(e^2*(m+1))*Int[(d+e*x)^(m+2)*Sin[a+b*x+c*x^2],x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[2*c*d-b*e,0] && LtQ[m,-1]
```

2. $\int (d+e\,x)^m \operatorname{Sin}\left[a+b\,x+c\,x^2\right] dx$ when $2\,c\,d-b\,e\neq 0$

1: $\int (d+e\,x) \operatorname{Sin}\left[a+b\,x+c\,x^2\right] dx$ when $2\,c\,d-b\,e\neq 0$

Rule: If $2\,c\,d-b\,e\neq 0$, then

$$\int (d+e\,x) \operatorname{Sin}\left[a+b\,x+c\,x^2\right] dx \longrightarrow -\frac{e \operatorname{Cos}\left[a+b\,x+c\,x^2\right]}{2\,c}+\frac{2\,c\,d-b\,e}{2\,c} \int \operatorname{Sin}\left[a+b\,x+c\,x^2\right] dx$$

Program code:

```
Int[(d_.+e_.*x_)*Sin[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  -e*Cos[a+b*x+c*x^2]/(2*c) +
  (2*c*d-b*e)/(2*c)*Int[Sin[a+b*x+c*x^2],x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[2*c*d-b*e,0]
```

```
Int[(d_.+e_.*x_)*Cos[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  e*Sin[a+b*x+c*x^2]/(2*c) +
  (2*c*d-b*e)/(2*c)*Int[Cos[a+b*x+c*x^2],x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[2*c*d-b*e,0]
```

2: $\int (d+e\,x)^m \,\text{Sin}\left[a+b\,x+c\,x^2\right] dx$ when $b\,e-2\,c\,d\neq 0 \,\wedge\, m>1$

Rule: If $b\,e-2\,c\,d\neq 0 \,\wedge\, m>1$**, then**

$$\int (d+e\,x)^m \,\text{Sin}\left[a+b\,x+c\,x^2\right] dx \,\longrightarrow$$
$$-\frac{e\,(d+e\,x)^{m-1}\,\text{Cos}\left[a+b\,x+c\,x^2\right]}{2\,c}-\frac{b\,e-2\,c\,d}{2\,c}\int (d+e\,x)^{m-1}\,\text{Sin}\left[a+b\,x+c\,x^2\right] dx+\frac{e^2\,(m-1)}{2\,c}\int (d+e\,x)^{m-2}\,\text{Cos}\left[a+b\,x+c\,x^2\right] dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*Sin[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  -e*(d+e*x)^(m-1)*Cos[a+b*x+c*x^2]/(2*c) -
  (b*e-2*c*d)/(2*c)*Int[(d+e*x)^(m-1)*Sin[a+b*x+c*x^2],x] +
  e^2*(m-1)/(2*c)*Int[(d+e*x)^(m-2)*Cos[a+b*x+c*x^2],x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b*e-2*c*d,0] && GtQ[m,1]
```

```
Int[(d_.+e_.*x_)^m_*Cos[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  e*(d+e*x)^(m-1)*Sin[a+b*x+c*x^2]/(2*c) -
  (b*e-2*c*d)/(2*c)*Int[(d+e*x)^(m-1)*Cos[a+b*x+c*x^2],x] -
  e^2*(m-1)/(2*c)*Int[(d+e*x)^(m-2)*Sin[a+b*x+c*x^2],x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b*e-2*c*d,0] && GtQ[m,1]
```

3: $\int (d+e\,x)^m \,\text{Sin}\left[a+b\,x+c\,x^2\right] dx$ when $b\,e-2\,c\,d\neq 0 \,\wedge\, m<-1$

Rule: If $b\,e-2\,c\,d\neq 0 \,\wedge\, m<-1$**, then**

$$\int (d+e\,x)^m \,\text{Sin}\left[a+b\,x+c\,x^2\right] dx \,\longrightarrow$$
$$\frac{(d+e\,x)^{m+1}\,\text{Sin}\left[a+b\,x+c\,x^2\right]}{e\,(m+1)}-\frac{b\,e-2\,c\,d}{e^2\,(m+1)}\int (d+e\,x)^{m+1}\,\text{Cos}\left[a+b\,x+c\,x^2\right] dx-\frac{2\,c}{e^2\,(m+1)}\int (d+e\,x)^{m+2}\,\text{Cos}\left[a+b\,x+c\,x^2\right] dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*Sin[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  (d+e*x)^(m+1)*Sin[a+b*x+c*x^2]/(e*(m+1)) -
  (b*e-2*c*d)/(e^2*(m+1))*Int[(d+e*x)^(m+1)*Cos[a+b*x+c*x^2],x] -
  2*c/(e^2*(m+1))*Int[(d+e*x)^(m+2)*Cos[a+b*x+c*x^2],x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b*e-2*c*d,0] && LtQ[m,-1]
```

```
Int[(d_.+e_.*x_)^m_*Cos[a_.+b_.*x_+c_.*x_^2],x_Symbol] :=
  (d+e*x)^(m+1)*Cos[a+b*x+c*x^2]/(e*(m+1)) +
  (b*e-2*c*d)/(e^2*(m+1))*Int[(d+e*x)^(m+1)*Sin[a+b*x+c*x^2],x] +
  2*c/(e^2*(m+1))*Int[(d+e*x)^(m+2)*Sin[a+b*x+c*x^2],x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b*e-2*c*d,0] && LtQ[m,-1]
```

2: $\int (d+e\,x)^m \operatorname{Sin}\left[a+b\,x+c\,x^2\right]^n \, dx$ when $n-1 \in \mathbb{Z}^+$

- **Derivation: Algebraic expansion**
- **Rule: If $n-1 \in \mathbb{Z}^+$, then**

$$\int (d+e\,x)^m \operatorname{Sin}\left[a+b\,x+c\,x^2\right]^n \, dx \longrightarrow \int (d+e\,x)^m \operatorname{TrigReduce}\left[\operatorname{Sin}\left[a+b\,x+c\,x^2\right]^n\right] \, dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_.*Sin[a_.+b_.*x_+c_.*x_^2]^n_,x_Symbol] :=
  Int[ExpandTrigReduce[(d+e*x)^m,Sin[a+b*x+c*x^2]^n,x],x] /;
FreeQ[{a,b,c,d,e,m},x] && IGtQ[n,1]
```

```
Int[(d_.+e_.*x_)^m_.*Cos[a_.+b_.*x_+c_.*x_^2]^n_,x_Symbol] :=
  Int[ExpandTrigReduce[(d+e*x)^m,Cos[a+b*x+c*x^2]^n,x],x] /;
FreeQ[{a,b,c,d,e,m},x] && IGtQ[n,1]
```

X: $\int (d+e\,x)^m \operatorname{Sin}\left[a+b\,x+c\,x^2\right]^n \, dx$

- **Rule:**

$$\int (d+e\,x)^m \operatorname{Sin}\left[a+b\,x+c\,x^2\right]^n \, dx \longrightarrow \int (d+e\,x)^m \operatorname{Sin}\left[a+b\,x+c\,x^2\right]^n \, dx$$

- **Program code:**

```
Int[(d_.+e_.*x_)^m_.*Sin[a_.+b_.*x_+c_.*x_^2]^n_.,x_Symbol] :=
  Unintegrable[(d+e*x)^m*Sin[a+b*x+c*x^2]^n,x] /;
FreeQ[{a,b,c,d,e,m,n},x]
```

```
Int[(d_.+e_.*x_)^m_.*Cos[a_.+b_.*x_+c_.*x_^2]^n_.,x_Symbol] :=
  Unintegrable[(d+e*x)^m*Cos[a+b*x+c*x^2]^n,x] /;
FreeQ[{a,b,c,d,e,m,n},x]
```

N: $\int u^m \,\text{Sin}[v]^n \,dx$ when $n \in \mathbb{Z}^+ \land u == d + e\,x \land v == a + b\,x + c\,x^2$

- **Derivation: Algebraic normalization**
- **Rule: If** $n \in \mathbb{Z}^+ \land u == d + e\,x \land v == a + b\,x + c\,x^2$**, then**

$$\int u^m \,\text{Sin}[v]^n \,dx \longrightarrow \int (d + e\,x)^m \,\text{Sin}\left[a + b\,x + c\,x^2\right]^n \,dx$$

- **Program code:**

```
Int[u_^m_.*Sin[v_]^n_.,x_Symbol] :=
  Int[ExpandToSum[u,x]^m*Sin[ExpandToSum[v,x]]^n,x] /;
FreeQ[m,x] && IGtQ[n,0] && LinearQ[u,x] && QuadraticQ[v,x] && Not[LinearMatchQ[u,x] && QuadraticMatchQ[v,x]]
```

```
Int[u_^m_.*Cos[v_]^n_.,x_Symbol] :=
  Int[ExpandToSum[u,x]^m*Cos[ExpandToSum[v,x]]^n,x] /;
FreeQ[m,x] && IGtQ[n,0] && LinearQ[u,x] && QuadraticQ[v,x] && Not[LinearMatchQ[u,x] && QuadraticMatchQ[v,x]]
```